Smart Trash Can
With Touch-Free Lid Operation
And Waste Level Indicator
Using Arduino

A Project

Presented to the Faculty of the

School of Management and Information Technology

Information Systems Program

De La Salle – College of Saint Benilde

Manila

In Partial Fulfillment

of the Requirements for the subject

ELECTIVE 1: EMBEDDED SYSTEMS

Submitted By:

JUICO, JOSE AURELIO N.

DR. JOSELITO EDUARD E. GOH

Professor

August 14, 2024

TABLE OF CONTENTS

		Page
Title Page	……………………………………….	1
Table of Contents	……………………………………….	2
Rubrics for Project Evaluation		3
A. Introduction	………….……………………………	4
B. Project Description	………………………………………	**4**
C. Scope and Limitation of the Project	………………………………………	**6**
D. Concept of Operation	..……….…………………………	**7**
E. Presentation of the Project Prototype	………….……………………………	**9**
F. Hardware Components	………….……………………………	**11**
G. Arduino Source Code	………….……………………………	**12**
Embedded System's Developer Profile	………….……………………………	**16**

A. INTRODUCTION

Automated devices with smart capabilities are increasingly being adopted in home and work environments. These devices improve user convenience, hygiene, and accessibility, as well as improve both workflow and energy efficiency, particularly in environments where repetitive tasks are common. In environments where proper sanitation is crucial, such as public places and healthcare facilities, and where users might have accessibility and mobility issues, smart, automated devices offer significant benefits.

Poor waste management practices can lead to the spread of diseases and pose health risks. Individuals with accessibility and mobility issues may also find it difficult to handle standard waste receptacles, which can lead to improper waste disposal.

B. PROJECT DESCRIPTION

This project aims to provide a proof-of-concept and working prototype of a smart trash can using common Arduino sensor modules. The concept can then be adapted to the production of commercial or industrial products.

This smart trash will be able to:

1. Improve sanitation and user hygiene by minimizing the need for physical contact with unclean surfaces.
2. Accommodate individuals with accessibility and mobility issues.
3. Improve waste disposal efficiency by displaying an easily identifiable indicator of the current amount of waste contained inside the trash can.

The project aims to address the urgency for proper sanitation and hygiene, especially in the context of past health-related events such as the COVID-19 pandemic. Minimizing physical contact with surfaces in public places has become crucial to reducing the transmission of viruses and disease-spreading microorganisms.

The smart trash can will be equipped with an automated lid mechanism and a waste level indicator. The design will incorporate two ultrasonic sensors, a servo motor, and an LCD to achieve the project's goals of enhancing user convenience and hygiene, providing user accessibility, and improving the overall sanitation of the area.

Automated Lid Operation:

The opening and closing of the lid will be managed by an ultrasonic sensor (HC-SR04) for proximity measurement and a servo motor (SG90) for the mechanical opening and closing of the lid. When an object, such as a hand, is detected near the sensor, the servo motor will activate to open the lid, eliminating the need for direct physical contact. This feature aims to address hygiene concerns associated with traditional, manually operated trash cans, reducing the spread of disease-carrying microorganisms. An LCD module (LCD1602) mounted on the side of the trash can will display whether the lid is opening or closing.

Waste Level Indicator:

An ultrasonic sensor (HY-SRF05) mounted inside the trash can will continuously monitor the waste fill level. When the automatic lid of the trash can is not in operation, the fill level of the contained waste will be displayed on the mounted LCD. The display allows users to easily assess whether the trash can already needs to be emptied. This functionality will contribute to maintaining a cleaner environment by preventing trash overflow and reducing the time the waste materials are present in the area.

Intended Users:

1. Individuals in Households
2. Individuals in Public Places
3. Individuals in Health-Risk Areas such as Healthcare Facilities
4. Individuals with Accessibility and Physical Mobility Issues

Functionality and Features:

1. Touch-free Operation
   An ultrasonic sensor on top of the trash can detects the presence of a hand or object and automatically opens the lid, eliminating the need for physical contact.
2. Waste Level Indicator
   An ultrasonic sensor measures the waste fill level inside the trash can and displays the corresponding indicator on the LCD mounted on the outside.
3. User Accessibility
   The touch-free functionality allows individuals who have physical issues to still be able to operate the device.

C. SCOPE AND LIMITATION OF THE PROJECT

Scope:

The scope includes developing a smart trash can prototype with the following features:

1. Touch-free lid opening and closing using ultrasonic sensors and a servo motor.
2. LCD displaying the opening and closing status of the trash can lid.
3. LCD displaying a visual indication of the waste fill level.

Limitations:

1. Ultrasonic Modules
   The sensors utilized are limited to the HC-SR04 and HY-SRF05 ultrasonic sensor modules. These sensors may have limitations in their accuracy and range as per their specifications.
2. Safety Sensor
   The project does not include the addition of safety sensors for preventing accidental injuries or misuse.
3. Child-Proofing
   The device does not include child-proofing features to prevent unintended usage.
4. Waste Compaction
   The device does not include a mechanism to compact waste items deposited inside the trash can, which can cause less accurate readings of the waste fill level.

Assumptions:

1. Proof of Concept
   The project is assumed to be only a proof of concept and is not designed as a complete end product.
2. Indoor Use
   The device is designed and intended to be used indoors with stable environmental conditions.
3. Limited Waste Types
   The device is assumed to handle only a limited range of waste items, such as everyday items available to the general public. Industrial and hazardous waste are not considered in the design.
4. User Operation
   The users are assumed to have a basic understanding of operating the device.

D. CONCEPT OF OPERATION



Fig 1. **3D Model with Closed Lid**

Fig. 1 is a 3D model of the device while the lid is closed. The mounting positions of the electronic components are shown. The SG90 Servo Motor handles the opening and closing of the lid. The HC-SR04 and the HY-SRF05 Ultrasonic Sensors handle the measurements for the object proximity and the waste fill level respectively. Lastly, the LCD1602 Liquid Crystal Display displays relevant information regarding the operation of the device.



Fig. 2 **High-Level Diagram**

Fig. 2 is a high-level diagram of the system including all of the components used and an indication of where the user interacts with the system. The power supply is externally located while all the other components are housed inside the trash can. The user interacts with the system through the HC-SR04 Ultrasonic Sensor.

E. PRESENTATION OF THE PROJECT PROTOTYPE



Fig. 3 **3D Model of the Touch-free Operation**

Fig. 3 presents a 3D model of the device in operation. When a hand or object is detected near the lid sensor, the servo motor is triggered to rotate to an angle that opens the trash can lid. During this process, the LCD screen displays 'Lid Opening,' and the waste fill level sensor temporarily pauses its readings.



Fig. 4 **Photo of LCD Displaying the Waste Level**

Fig. 4 is a photograph of the LCD when the trash can lid is closed. The waste fill level sensor, positioned inside the trash can, measures the height of the deposited waste. The second line of the LCD module shows the current waste fill level, represented by a bar graph that visually depicts the fullness of the trash can.

F. HARDWARE COMPONENTS



Fig. 5. **Circuit Diagram 1**

Fig. 5 shows the circuit diagram illustrating how the components are interconnected. All components (HC-SR04, HY-SRF05, SG90, LCD1602) are connected in parallel via a mini breadboard and are powered by the 5V output from the Arduino Uno R3. Each component's signal pin is wired to a dedicated digital I/O pin on the Arduino for independent control.



Fig. 6. **Schematic Diagram 1**

Fig. 6 shows the schematic diagram illustrating the electrical connections between the components. The diagram depicts how each component is connected to the Arduino Uno R3, with labeled power and ground connections. Signal connections are shown as routed to specific digital I/O pins on the Arduino.

G. ARDUINO SOURCE CODE

```
#include <Servo.h>
#include <NewPing.h>
#include <Wire.h>
#include <LiquidCrystal_I2C.h>

#define SERVO_PIN 3
Servo servo1;
float servoAngle = 90.0;

// Lid operation sensor
#define LID_TRIGGER_PIN 9
#define LID_ECHO_PIN 10
#define LID_MAX_DISTANCE 20

// Waste fill segment sensor
#define FILL_TRIGGER_PIN 11
#define FILL_ECHO_PIN 12
#define FILL_MAX_DISTANCE 26

NewPing sonar1(LID_TRIGGER_PIN, LID_ECHO_PIN, LID_MAX_DISTANCE);
NewPing sonar2(FILL_TRIGGER_PIN, FILL_ECHO_PIN, FILL_MAX_DISTANCE);

LiquidCrystal_I2C lcd(0x27, 16, 2);

// Variables for multitasking
unsigned long lidOpenTime = 0;
unsigned long lidMoveStartTime = 0;
unsigned long previousLidTime = 0;
unsigned long previousWasteTime = 0;

const long lidInterval = 50;
const long lidMoveDuration = 1500;
const long wasteInterval = 500;
const int servoSteps = 50;
```

```
// Keeps track of the lid state
bool isLidOpen = false;
bool isLidMoving = false;

// Stores the last valid waste reading
int lastValidWaste = FILL_MAX_DISTANCE;

void setup() {
  Wire.begin();
  lcd.init();
  lcd.backlight();

  pinMode(SERVO_PIN, OUTPUT);
  servo1.attach(SERVO_PIN);
  servo1.write(servoAngle);

  // Static text on LCD line 1
  lcd.setCursor(0, 0);
  lcd.print("Smart Trash Can");
}

void moveLid(float startAngle, float endAngle, unsigned long startTime) {
  unsigned long currentTime = millis();
  unsigned long elapsedTime = currentTime - startTime;

  if (elapsedTime <= lidMoveDuration) {
    float servoStepSize = (endAngle - startAngle) / float(servoSteps);
     float angle = startAngle + (servoStepSize * (elapsedTime / (lidMoveDuration /
float(servoSteps))));
    servo1.write(angle);
  } else {
    servo1.write(endAngle);
    isLidMoving = false;
    servoAngle = endAngle;
  }
}

void loop() {
  unsigned long currentTime = millis();

  // Checks for presence of object near the lid
  if (currentTime - previousLidTime >= lidInterval) {
    previousLidTime = currentTime;
```

```
  int distance = sonar1.ping_cm();

  if (!isLidOpen && !isLidMoving && distance >= 2 && distance <= LID_MAX_DISTANCE) {
    lcd.setCursor(0, 1);
    lcd.print("Lid Opening      ");
    isLidMoving = true;
    lidMoveStartTime = currentTime;
    isLidOpen = true;
    lidOpenTime = currentTime;
  } else if (isLidOpen && currentTime - lidOpenTime >= 5000) {
    lcd.setCursor(0, 1);
    lcd.print("Lid Closing      ");
    isLidMoving = true;
    lidMoveStartTime = currentTime;
    isLidOpen = false;
  }
}

// Checks waste fill level every set interval
if (currentTime - previousWasteTime >= wasteInterval) {
  previousWasteTime = currentTime;

  int waste = sonar2.ping_cm();

  // Uses last valid reading if current reading is invalid (0)
  if (waste == 0) {
    waste = lastValidWaste;
  } else {
    lastValidWaste = waste;
  }

  // Variable for how many bars to display
  int bars = 0;

  if (waste < FILL_MAX_DISTANCE && waste >= 2) {
    // Divides the max distance measured into 5 equal parts
    float segment = (FILL_MAX_DISTANCE - 2.0) / 5.0;

    if (waste < FILL_MAX_DISTANCE) bars = 1;
    if (waste < FILL_MAX_DISTANCE - segment) bars = 2;
    if (waste < FILL_MAX_DISTANCE - 2 * segment) bars = 3;
    if (waste < FILL_MAX_DISTANCE - 3 * segment) bars = 4;
    if (waste < FILL_MAX_DISTANCE - 4 * segment) bars = 5;
  }
```

```
    // Display waste level only when the lid is not moving
    if (!isLidMoving && !isLidOpen) {
      lcd.setCursor(0, 1);
      lcd.print("Waste Lvl:");

      for (int i = 0; i < bars; i++) {
        lcd.write(0xFF);  // Block character for bars
      }
      for (int i = bars; i < 5; i++) {
        lcd.print(" ");  // Empty space for unused segments
      }
    }
  }

  // Determines whether to open or close lid
  if (isLidMoving) {
    if (isLidOpen) {
      moveLid(servoAngle, 175.0, lidMoveStartTime);
    } else {
      moveLid(servoAngle, 90.0, lidMoveStartTime);
    }
  }
}
```
